Chapter 14 Differentiation in Several Variables

Useful Tip: If you are reading the electronic version of this publication formatted as a *Mathematica* Notebook, then it is possible to view 3-D plots generated by *Mathematica* from different perspectives. First, place your screen cursor over the plot. Then drag the mouse while pressing down on the left mouse button to rotate the plot.

14.1 Functions of Two or More Variables

Students should read Section 14.1 of Rogawski's *Calculus* [1] for a detailed discussion of the material presented in this section.

14.1.1 Plotting Level Curves using ContourPlot

We begin with plotting level curves $f(x, y) = c$ of a function of two variables. The command to plot level curves is **ContourPlot[f,{x,a,b},{y,c,d}]**.

Most of the options for **ContourPlot** are the same as those for **Plot**. In the following example, we consider the option **ImageSize**.

Example 14.1. Plot the level curves of $f(x, y) = x^2 + x\,y - y^2$.

Solution: Let us first plot the level curves using the default settings of *Mathematica*.

In[154]:=
```
Clear[x, y, f]
f[x_, y_] := x^2 + x y - y^2
```

In[156]:=
```
ContourPlot[f[x, y], {x, -5, 5}, {y, -5, 5}, ImageSize → {250}]
```

Out[156]=



To get the level curves on the *xy*-plane without the shading, the colors, and the frame, but with the coordinate axes, we use the following options of **ContourPlot**.

In[157]:=
```
ContourPlot[f[x, y], {x, -5, 5}, {y, -5, 5}, Frame → False,
  Axes → True, ContourShading → False, ImageSize → {250}]
```

Out[157]=



Contours is an option of **ContourPlot** that can be used in two different ways: **Contour→n** displays **n** equally spaced contour curves while **Contour→list** plots level curves $f(x, y) = c$ where c is an element of the list **list**.

To plot 15 level curves, we evaluate

In[158]:=
```
ContourPlot[f[x, y], {x, -1, 1}, {y, -1, 1}, Contours → 15, ImageSize → {250} ]
```

Out[158]=



Here is an example when **list** = {−10, −5, −2, −1, 0, 1, 2, 5, 10}.

In[159]:=
```
ContourPlot[f[x, y], {x, -5, 5}, {y, -5, 5},
 Contours → {-10, -5, -2, -1, 0, 1, 2, 5, 10}, ImageSize → {250}]
```

Out[159]=



14.1.2 Plotting Surfaces using Plot3D

Plot3D is the three-dimensional analog of the **Plot** command. Given an expression in two variables and the domain for the variables, **Plot3D** produces a surface plot.

The basic syntax to plot the graph of a function of two variables is **Plot3D[f,{x, a, b},{y, c, d}]**, where **f** is a function of x and y with $a \leq x \leq b$ and $c \leq y \leq d$.

The command to plot the graphs of two or more functions on the same coordinate axes is **Plot3D[{f, g, h, ... }, {x, a, b}, {y, c, d}]**, where **f, g, h, ...** are the functions to be plotted.

We will begin with the default settings of plotting a graph of a function of two variables.

Example 14.2. Plot $f(x, y) = \sin(x - \cos y)$.

Solution:

In[160]:=
```
Plot3D[Sin[x - Cos[y]], {x, -3, 3}, {y, -3, 3}]
```

Out[160]=



Example 14.3. Plot the graphs of $f(x, y) = 3x + 4y - 3$ and $g(x, y) = 10\sin(xy)$ on the same axes.

Solution: We will use red color for the graph of f and blue for that of g. This is given using the option **PlotStyle**.

In[161]:=
```
Plot3D[{3 x + 4 y - 3, 10 Sin[x y]}, {x, -5, 5},
 {y, -5, 5}, PlotStyle → {Green, Blue}, ImageSize → {250}]
```

Out[161]=



NOTE: One of the most significant improvements of *Mathematica 7.0* over the previous editions is its graphics capability. **Plot3D** has many options. Evaluate the command **Options[Plot3D]** to see the many options you have to plot a nice graph.

We will discuss some of these options below.

ViewPoint

In *Mathematica 7.0*, we can rotate the graph of a function of two variables by simply clicking on the graph and dragging the mouse around to see the graph from any point of view.

The option **ViewPoint** specifies the point in space from which the observer looks at a graphics object. The syntax for choosing a view point of a surface is **Plot3D[f[x, y], {x, a, b}, {y, c, d}, ViewPoint→{A, B, C}]**. The default value for **{A, B, C}** is {1.3,-2.4,2.0}. This may be changed by entering values directly.

To view a graph from directly in front {0, −2, 0}; in front and up {0, −2, 2}; in front and down {0, −2, −2}; left hand corner { −2, −2, 0}; directly above {0, 0, 2}.

Plot3D[f[x, y], {x, a, b}, {y, c, d}, ViewPoint → view] produces a plot viewed from **view**. The possible values of **view** are **Above** (along positive z-axis), **Below** (along negative z-axis), **Front** (along negative y-axis), **Back** (along positive y-axis), **Left** (along the negative x-axis), and **Right** (along the positive x-axis**)**.

Example 14.4. Plot $f(x, y) = \cos x \sin y$ using **ViewPoint** option to view the graph from various view points.

Solution: We leave it to the reader to navigate all of the above choices. We will consider a few of them.

In[162]:=
```
Clear[f]
f[x_, y_] = Cos[x] Sin[y]
```

Out[163]= Cos[x] Sin[y]

Here is a plot of the graph using the default setting for **ViewPoint**:

```
In[164]:= Plot3D[ f[x, y], {x, -2 Pi, 2 Pi}, {y, -2 Pi, 2 Pi} , PlotRange → All,
           ImageSize → {250} ]
```

Out[164]=



View from directly in front:

```
In[165]:= Plot3D[ f[x, y], {x, -2 Pi, 2 Pi}, {y, -2 Pi, 2 Pi}, ViewPoint → Front ,
           PlotRange → All , ImageSize → {250}]
```

Out[165]=



View from in front and up:

```
In[166]:= Plot3D[ f[x, y], {x, -2 Pi, 2 Pi}, {y, -2 Pi, 2 Pi}, ViewPoint → {0, -2, 2},
           PlotRange → All, ImageSize → {250}]
```

Out[166]=



View from in front and down:

In[167]:=
```
Plot3D[f[x, y], {x, -2 Pi, 2 Pi}, {y, -2 Pi, 2 Pi}, ViewPoint → {0, -2, -2},
  PlotRange → All, ImageSize → {250}]
```

Out[167]=



View from directly above:

In[168]:=
```
Plot3D[f[x, y], {x, -2 Pi, 2 Pi}, {y, -2 Pi, 2 Pi}, ViewPoint → Above,
  PlotRange → All, Ticks → {Automatic, Automatic, {-1, 0, 1}},
  ImageSize → {250}]
```

Out[168]=



View from the right:

In[169]:=
```
Plot3D[f[x, y], {x, -2 Pi, 2 Pi}, {y, -2 Pi, 2 Pi}, ViewPoint → Right,
  PlotRange → All, ImageSize → {250}]
```

Out[169]=



NOTE: As we pointed out earlier, we can also select different viewpoints by clicking on the graph and dragging the mouse around until we get the desired viewpoint.

Mesh, **MeshStyle, MeshShading**

The option **Mesh** specifies the type of mesh that should be drawn.

The option **MeshStyle** specifies the style in which a mesh should be drawn.

The option **MeshShading** is an option for specifying a list of colors to be used between mesh divisions.

We illustrate some uses of these options in the example below.

Example 14.5. Plot $f(x, y) = \cos x \sin y$ using various options involving **Mesh**.

Solution:

```
In[170]:= Clear[f]
          f[x_, y_] = Cos[x] Sin[y]
Out[171]= Cos[x] Sin[y]
```

To plot a graph without a mesh we use the setting **Mesh→None**.

```
In[172]:= Plot3D[ f[x, y], {x, -2 Pi, 2 Pi}, {y, -2 Pi, 2 Pi} , Mesh → None,
           ImageSize → {250} ]
```

Out[172]=



Mesh→n plots a surface with only $n \times n$ meshes.

In[173]:=
```
Plot3D[ f[x, y], {x, -2 Pi, 2 Pi}, {y, -2 Pi, 2 Pi} , Mesh → 8 ,
 ImageSize → {250}]
```

Out[173]=



We can choose the color of the mesh using **MeshStyle**.

In[174]:=
```
Plot3D[ f[x, y], {x, -2 Pi, 2 Pi}, {y, -2 Pi, 2 Pi} , MeshStyle →
  {Red, Black}, ImageSize → {250} ]
```

Out[174]=



Here is another use of **MeshStyle**:

In[175]:=
```
Plot3D[ f[x, y], {x, -2 Pi, 2 Pi}, {y, -2 Pi, 2 Pi} , MeshStyle →
  {Dashing[0.01], None}, ImageSize → {250} ]
```

Out[175]=



To display a plot with selected colors between meshes we use **MeshShading**:

```
In[176]:= Plot3D[f[x, y], {x, -2 Pi, 2 Pi}, {y, -2 Pi, 2 Pi},
  MeshShading → {{Blue, Red, White}, {Purple, Green, Black}}, ImageSize → {250} ]
```

Out[176]=



Here is a neat example in *Mathematica* 7.0:

```
In[177]:= Plot3D[(x^2 - y^2) / (x^2 + y^2)^2, {x, -1.5, 1.5}, {y, -1.5, 1.5},
  BoxRatios → Automatic, PlotPoints → 25, MeshFunctions → {#3 &},
  MeshStyle → Purple, MeshShading → {None, Green, None, Yellow}, ImageSize → {250}]
```

Out[177]=



BoxRatios

The option **BoxRatios** specifies the ratio of the lengths of the sides of the box. This is analogous to specifying the **AspectRatio** of a two-dimensional plot. For **Plot3D**, the default setting is **BoxRatios→Automatic.**

Example 14.6. Plot $f(x, y) = e^{1-x^2-y^2}$ using the **BoxRatio** option.

Solution:

In[178]:=
```
Clear[f]
f[x_, y_] = E^(1-x^2-y^2)
```

Out[179]= $e^{1-x^2-y^2}$

In[180]:=
```
Plot3D[ f[x, y], {x, -2, 2}, {y, -2, 2}, ImageSize → {250}]
```

Out[180]=



In[181]:=
```
Plot3D[ f[x, y], {x, -2, 2}, {y, -2, 2}, BoxRatios → {1, 1, 0.62`},
 ImageSize → {250}]
```

Out[181]=



AxesLabel

The option **AxesLabel** is a command used to label the axes in plotting.

Example 14.7. Plot $f(x, y) = \sqrt{9 - x^2 - y^2}$ using the **AxesLabel** option.

Solution:

In[182]:=
```
Clear[f]
f[x_, y_] = Sqrt[9 - x^2 - y^2]
```

Out[183]= $\sqrt{9 - x^2 - y^2}$

In[184]:=
```
Plot3D[f[x, y], {x, -3, 3}, {y, -3, 3}, AxesLabel → {"x ", "y ", "z "},
 ImageSize → {250}, ImagePadding → {{15, 15}, {15, 15}}]
```

Out[184]=



NOTE: To label a graph, use the **PlotLabel** option as shown following:

In[185]:=
```
Plot3D[f[x, y], {x, -3, 3}, {y, -3, 3}, AxesLabel → {"x ", "y ", "z "},
 PlotLabel → "Upper hemisphere", BoxRatios → Automatic, ImageSize → {250},
 ImagePadding → {{15, 15}, {15, 25}}]
```

Out[185]=



ColorFunction

The option **ColorFunction** specifies a function to apply to the values of the function being plotted to determine the color to use for a particular region on the *xy*-plane. It is an option for **Plot3D**, **ListPlot3D**, **DensityPlot**, and **ContourPlot**. The default setting for **ColorFunction** is **ColorFunction→Automatic**. **ColorFunction→Hue** yields a range of colors.

Example 14.8. Plot $f(x, y) = \sin(x^2 + y^2) + e^{1-x^2-y^2}$ in various colors using the **ColorFunction** option.

Solution:

In[186]:=
```
Clear[f, x, y]
f[x_, y_] = Sin[x^2 + y^2] + E^(1-x^2-y^2)
```

Out[187]= $e^{1-x^2-y^2} + \mathrm{Sin}\left[x^2 + y^2\right]$

In[188]:=
```
Plot3D[f[x, y], {x, -Pi, Pi}, {y, -Pi, Pi}, ColorFunction → Hue,
 ImageSize → {250}]
```

Out[188]=



Here are other ways to use **ColorFunction**.

In[189]:=
```
Plot3D[f[x, y], {x, -Pi, Pi}, {y, -Pi, Pi}, ColorFunction → "Rainbow",
 ImageSize → {250} ]
```

Out[189]=



In[190]:=
```
Plot3D[f[x, y], {x, -Pi, Pi}, {y, -Pi, Pi}, ColorFunction →
  "BlueGreenYellow", ImageSize → {250}]
```

Out[190]=



NOTE: We can use **PlotStyle** option to select color for graphs. The plot below uses this option.

In[191]:=
```
Plot3D[f[x, y], {x, -Pi, Pi}, {y, -Pi, Pi}, PlotStyle → Yellow,
 ImageSize → {250} ]
```

Out[191]=



RegionFunction

The option **RegionFunction** specifies the region to include in the plot drawn.

Example 14.9. Plot $f(x, y) = \begin{cases} 10\sin(3x - y), & \text{if } x^2 + y^2 < 4; \\ x^2 + y^2 - 5, & \text{otherwise} \end{cases}$.

Solution: We will use the command **RegionFunction** to specify the domain $x^2 + y^2 < 4$ as follows. Note that we have used **Show** to display the graphs.

In[192]:=
```
Clear[plot1, plot2]
plot1 = Plot3D[10 Sin[3 x - y], {x, -4, 4}, {y, -4, 4}, PlotStyle → Blue,
    RegionFunction → Function[{x, y, z}, x^2 + y^2 < 4]];
plot2 = Plot3D[ x^2 + y^2 - 5, {x, -4, 4}, {y, -4, 4}, PlotStyle → Red,
    RegionFunction → Function[{x, y, z}, x^2 + y^2 ≥ 4]];
Show[plot1, plot2, ImageSize → {250}]
```

Out[195]=



If we want to focus on a particular part of a surface defined by a function, we can use the option **RegionFunction**. The following example shows this point.

Example 14.10. Plot the graph of $f(x, y) = x^2 - 3xy - 2y^2$ and show the portion of the surface direclty above the unit circle centered at the origin.

Solution: We will use the option **ViewPoint**.

In[196]:=
```
Clear[plot1, plot2, f, x, y]
f[x_, y_] = x^2 - 3 x y - 2 y^2
plot1 = Plot3D[f[x, y], {x, -4, 4}, {y, -4, 4}, PlotStyle → Blue,
    RegionFunction → Function[{x, y, z}, x^2 + y^2 < 1 ]];
plot2 = Plot3D[f[x, y] , {x, -4, 4}, {y, -4, 4}, PlotStyle → Red,
   RegionFunction → Function[{x, y, z}, x^2 + y^2 > 1 ]];
Show[plot1, plot2 , ViewPoint → Front, ImageSize → {250} ]
```

Out[197]= $x^2 - 3\,x\,y - 2\,y^2$

Out[200]=



■ 14.1.3 Plotting Parametric Surfaces using ParametricPlot3D

ParametricPlot3D is a direct analog of **ParametricPlot.** Depending on the input, **ParametricPlot3D** produces a space curve or a surface. **ParametricPlot3D[{f, g, h}, {t, a, b }]** produces a three-dimensional space curve parametrized by the variable **t**, which runs from **a** to **b**. **ParametricPlot3D[{f, g, h}, {t, a, b },{u, c, d}]** produces a two-dimensional surface parametrized by **t**

and **u**. Options are given to **ParametricPlot3D** the same way as for **Plot3D**. Most of the options are the same.

Example 14.11. Plot the curve that is parametrized by $x = \sin t$, $y = \cos t$ and $z = t/3$ with $0 \leq t \leq 2\pi$.

Solution:

```
In[201]:= ParametricPlot3D[{Sin[t], Cos[t], t/3}, {t, 0, 2 π}, ImageSize → {250},
  ImagePadding → {{15, 15}, {15, 15}}]
```

Out[201]=



Example 14.12. Plot the surface that is parametrized by $x = u \cos u\,(4 + \cos(u + v))$, $y = u \sin u\,(4 + \cos(u + v))$, and $z = u \sin(u + v)$.

Solution:

```
In[202]:= ParametricPlot3D[{u Cos[u] (4 + Cos[u + v]), u Sin[u] (4 + Cos[u + v]), u Sin[u + v]},
  {u, 0, 4 π}, {v, 0, 2 π}, ImageSize → {250}]
```

Out[202]=



■ 14.1.4 Plotting Level Surfaces using ContourPlot3D

ContourPlot3D is the command used to plot level surfaces of functions of three variables. Its syntax is **ContourPlot3D[f,{x,a,b}, {y,c,d},{z,e,f}]**. Most of the Options for **ContourPlot3D** are the same as those of **Plot3D**. Below we will consider the option **Contours** of **ContourPlot3D**.

Example 14.13. Plot level surfaces of $f(x, y, z) = x^2 + y^2 + z^2$.

In[203]:=
```
Clear[x, y, z, f]
f[x_, y_, z_] = x^2 + y^2 + z^2
ContourPlot3D[f[x, y, z], {x, -3, 3}, {y, -3, 3}, {z, -3, 3}, ImageSize → {250}]
```

Out[204]= $x^2 + y^2 + z^2$

Out[205]=



The following displays five (5) equally spaced contour surfaces of f.

In[206]:=
```
ContourPlot3D[f[x, y, z], {x, -3, 3}, {y, -3, 3}, {z, -3, 3},
 Contours → 5, ImageSize → {250}]
```

Out[206]=



The following displays three level surfaces $f(x, y, z) = c$, where $c = 1, 4, 9$.

In[207]:=
```
ContourPlot3D[f[x, y, z], {x, -3, 3}, {y, -3, 3}, {z, -3, 3},
 Contours → {1, 4, 9}, ImageSize → {250}]
```

Out[207]=



Notice that we only see one sphere. The other two are enclosed in the sphere of radius 3 corresponding to $c = 9$. One way to remedy this is to plot the level surfaces one by one. For this we use the **GraphicsArray** command. First, let us define the level surfaces as function of c:

In[208]:=
```
Clear[c, plot]
plot[c_] := ContourPlot3D[f[x, y, z], {x, -3, 3}, {y, -3, 3}, {z, -3, 3},
   Contours → {c}]
```

Here are the three level surfaces corresponding to $c = 1, 4, 9$.

In[210]:=
```
Show[GraphicsArray[{plot[1], plot[4], plot[9]}]]
```

GraphicsArray::obs : GraphicsArray is obsolete. Switching to GraphicsGrid. ≫

Out[210]=



Exercises

In Exercises 1 through 4, plot the level curves and the graphs of the given functions.

1. $f(x, y) = x y^5 - x^5 y$ for $-10 \le x \le 10$, $-10 \le y \le 10$
2. $f(x, y) = \frac{x^2+2y}{1+x^2+y^2}$ for $-10 \le x \le 10$, $-10 \le y \le 10$
3. $f(x, y) = (\sin y) e^{\cos x}$ for $-2\pi \le x \le 2\pi$, $-2\pi \le y \le 2\pi$
4. $f(x, y) = \sin(x + \sin(y))$ for $-4\pi \le x \le 4\pi$, $-4\pi \le y \le 4\pi$

In Exercises 5 through 7, use at least two nondefault options to plot the given functions.

5. $f(x, y) = \sin(x - 2y) e^{1/(y-x)}$ for $-2\pi \le x \le 2\pi$, $-2\pi \le y \le 2\pi$

6. $f(x, y) = 4 - 3\,|\,x\,| - 2\,|\,y\,|$ for $-10 \le x \le 10$, $-10 \le y \le 10$
7. $f(x, y) = \tanh^{-1}(x/y)$ for $-5 \le x \le 5$, $-5 \le y \le 5$

8. Plot $f(x, y) = \begin{cases} x^2 + y^2 - 4 & \text{if } x^2 + y^2 < 4 \\ 4 - x^2 + 3\,y^2 & \text{otherwise} \end{cases}$

9. Plot the portion of the *helicoid (spiral ramp)* that is defined by:
$x = u\cos v$, $y = u\sin v$, $z = v$ for $0 \le u \le 3$ and $-2\pi \le v \le 2\pi$

10. Use **ContourPlot3D** to plot the level surfaces of the function $f(x, y, z) = 9 - x^2 - y^2 - z^2$.

■ 14.2 Limits and Continuity

Students should read Section 14.2 of Rogawski's *Calculus* [1] for a detailed discussion of the material presented in this section.

■ 14.2.1 Limits

If $f(x, y)$ is a function of x and y, and if the domain of f contains a circle around the point (a, b), we say that the limit of f at (a, b) is L if and only if $f(x, y)$ can be arbitrarily close to L for all (x, y) arbitrarily close (a, b).

More precisely, for a given $\epsilon > 0$, there exists a $\delta > 0$ such that for every (x, y) is in the domain of f,

$$0 < \sqrt{(x-a)^2 + (y-b)^2} < \delta \implies |f(x, y) - L| < \epsilon$$

If this is the case, we write

$$\lim_{(x,y)\to(a,b)} f(x, y) = L$$

The **Limit** command of *Mathematica* is restricted to functions of one variable. However, we can use it twice to find the limit of function of two variables provided the limit exists.

Example 14.14. Find $\lim_{(x,y)\to(3,4)} \left(x^2 + y^2\right)$.

Solution: We can easliy determine that the limit exists. We can find the limit by evaluating

In[211]:=
```
Limit[Limit[x^2 + y^2, x → 3], y → 4]
```

Out[211]=
```
25
```

The plot following confirms this.

In[212]:=
```
Clear[plot1, plot2]
plot1 = Plot3D[x^2 + y^2, {x, 1, 4}, {y, 3, 5}];
plot2 = Graphics3D[{Red, PointSize[.025], Point[{3, 4, 25}]}];
Show[plot1, plot2, ImageSize → {250}, ImagePadding → {{15, 15}, {15, 15}}]
```

Out[215]=



Example 14.15. Find $\lim_{(x,y)\to(4,1)} \frac{3x+y^2}{x-4y}$.

Solution: We will evaluate the limit in two different orders. The limit in which we use limit with x first and then with y is

In[216]:=
```
Clear[f, x, y]
f[x_, y_] = (3 x + y^2)/(x - 4 y)
```

Out[217]= $\frac{3\,x + y^2}{x - 4\,y}$

The limit in which we use limit with x first and then with y is

In[218]:=
```
Limit[Limit[f[x, y], x → 4], y → 1]
```

Out[218]= $-\infty$

The limit in which we use limit with y first and then with x is

In[219]:=
```
Limit[Limit[f[x, y], y → 1], x → 4]
```

Out[219]= ∞

Here is the plot of the graph near the point (4, 1). Observe that the graph of the function is in green and the point (4, 1, 0) is in red. For a better comaprison, we have colored the xy-plane light blue. You may need to rotate the graph to see the point (4, 1, 0) on the xy-plane and see how the graph behaves when (x, y) is close to (4, 1).

In[220]:=
```
Clear[plot1, plot2]
plot1 = Plot3D[{f[x, y], 0}, {x, 3, 5},
   {y, 0, 2}, PlotStyle → {Green, LightBlue}, PlotPoints → 100];
plot2 = Graphics3D[{Red, PointSize[.025], Point[{4, 1, 0}]}];
Show[plot1, plot2, ImageSize → {250},
 ImagePadding → {{15, 15}, {15, 15}}]
```

Out[223]=



Here is the animation with *x* as the animation parameter.

Important Note: If you are reading the printed version of this publication, then you will not be able to view any of the animations generated from the **Animate** command in this chapter. If you are reading the electronic version of this publication formatted as a *Mathematica* Notebook, then evaluate each **Animate** command to view the corresponding animation. Just click on the arrow button to start the animation. To control the animation just click at various points on the sliding bar or else manually drag the bar.

In[224]:=
```
Animate[Plot[f[x, y], {y, 0, 3}, PlotRange → {-20, 20}], {x, 3, 5}]
```

Out[224]=



Example 14.16. Find $\lim_{(x,y)\to(0,0)} \frac{\sin x \sin y}{x\, y}$.

Solution: We will evaluate the limit in two different orders.

In[225]:=
```
Clear[f, x, y]
f[x_, y_] = Sin[x y] / (x y)
```

Out[226]= $\frac{\text{Sin}[x\, y]}{x\, y}$

In[227]:=
```
Limit[Limit[f[x, y], x → 0], y → 0]
```

Out[227]= 1

In[228]:=
```
Limit[Limit[f[x, y], y → 0], x → 0]
```

Out[228]= 1

Here is the plot of the graph and the point (0, 0, 1).

In[229]:=
```
Clear[plot1, plot2]
plot1 = Plot3D[f[x, y], {x, -1, 1}, {y, -1, 1}, PlotStyle → Green];
plot2 = Graphics3D[{Red, PointSize[.02], Point[{0, 0, 1}]}];
Show[plot1, plot2, ImageSize → {250}]
```

Out[232]=



If we rotate this graph to a suitable position, we notice that the limit exists. Here are animations with *x* and *y* as animation parameters, respectively.

In[233]:=
```
Animate[Plot[f[x, y], {x, -2, 2}, PlotRange → {0, 1}], {y, -2, 2}]
```

Out[233]=



In[234]:=
```
Animate[Plot[f[x, y], {y, -2, 2}, PlotRange → {0, 1}], {x, -2, 2}]
```

Out[234]=



Example 14.17. Find $\lim_{(x,y)\to(0,0)} x \ln y$.

Solution:

In[235]:=
```
Clear[f, x, y]
f[x_, y_] = x Log[y]
```

Out[236]= `x Log[y]`

In[237]:=
```
Limit[Limit[f[x, y], x → 0], y → 0]
```

Out[237]= `0`

In[238]:=
```
Limit[Limit[f[x, y], y → 0], x → 0]
```

Out[238]= `Indeterminate`

In[239]:=
```
Clear[plot1, plot2]
plot1 =
  Plot3D[{f[x, y], 0}, {x, -1, 1}, {y, -1, 1}, PlotStyle → { Green, LightBlue}];
plot2 = Graphics3D[{Red, PointSize[.025], Point[{0, 0, 0}]}];
Show[plot1, plot2, ImageSize → {250},
 ImagePadding → {{15, 15}, {15, 15}}]
```

Out[242]=



Here is the animation with *x* as the animation parameter.

In[243]:=
```
Animate[Plot[f[x, y], {y, -2, 2}, PlotRange → {-10, 10}], {x, -2, 2}]
```

Out[243]=



Example 14.18. Consider the function $f(x, y) = \frac{x y^2}{x^2+y^4}$. Show that $\lim_{(x,y)\to(0,0)} f(x, y)$ does not exist.

Solution:

In[244]:=
```
Clear[f, x, y]
f[x_, y_] = x y^2/(x^2 + y^4)
```

Out[245]= $\frac{x\,y^2}{x^2 + y^4}$

In[246]:=
```
Limit[Limit[f[x, y], x → 0], y → 0]
```

Out[246]= 0

In[247]:=
```
Limit[Limit[f[x, y], y → 0], x → 0]
```

Out[247]= 0

In[248]:=
```
Limit[Limit[f[x, y], y → m x], x → 0]
```

Out[248]= 0

However, note that the limit along the curve $y = \sqrt{x}$ is

In[249]:=
```
Limit[Limit[f[x, y], y → √x], x → 0]
```

Out[249]= $\frac{1}{2}$

Hence, the limit does not exist. Here is the plot of the function:

In[250]:=
```
Plot3D[f[x, y], {x, -1, 1}, {y, -1, 1}, ImageSize → {250}]
```

Out[250]=



■ 14.2.2 Continiuty

Recall that a function f of two variables x and y is continuous at the point (a, b) if and only if $\lim_{(x,y)\to(a,b)} f(x, y) = f(a, b)$.

Example 14.19. Let $f(x, y) = \begin{cases} 1 - x^2 - y^2, & \text{if } x^2 + y^2 < 1 \\ 0, & \text{if } x^2 + y^2 \geq 1 \end{cases}$. Is f continuous?

Solution: Clearly, f is continuous at all points inside and outside the circle of radius 1. To check continuity on the unit circle, we let $x = r\cos t$ and $y = r\sin t$. We then let $r \to 1$.

In[251]:=
```
Clear[x, y, r, s, t, f]
f[x_, y_] = 1 - Sqrt[x^2 + y^2]
```

Out[252]= $1 - \sqrt{x^2 + y^2}$

In[253]:=
```
x = r Cos[t]
y = r Sin[t]
```

Out[253]= r Cos[t]

Out[254]= r Sin[t]

In[255]:=
```
Simplify[f[x, y]]
```

Out[255]= $1 - \sqrt{r^2}$

In[256]:=
```
Limit[f[x, y], r → 1]
```

Out[256]= 0

The command below evaluates *f* on the circle.

In[257]:=
```
Simplify[f[x, y] /. r → 1]
```

Out[257]= 0

Thus, the limit and the value of *f* are equal at all points on the unit circle. Hence, *f* is continuous everywhere. Here is the graph.

In[258]:=
```
Clear[plot1, plot2]
plot1 = Plot3D[f[x, y], {x, -5, 5}, {y, -5, 5}, PlotStyle → Red,
   RegionFunction → Function[{x, y, z}, x^2 + y^2 < 1], Mesh → None];
plot2 = Plot3D[ 0, {x, -5, 5}, {y, -5, 5}, PlotStyle → LightBlue,
   RegionFunction → Function[{x, y, z}, x^2 + y^2 ≥ 1], Mesh → None];
Show[plot1, plot2, ImageSize → {250}]
```

Out[261]=


Exercises

In Exercises 1 through 4, find the limit, if it exists.

1. $\lim_{(x,y)\to(1,-1)} \left(2\, x^2\, y + x\, y^2\right)$ 2. $\lim_{(x,y)\to(1,1)} \frac{3\, x^2 + y^2}{x^2 - y}$

3. $\lim_{(x,y)\to(0,0)} \frac{\tan x \sin y}{x\,y}$

4. $\lim_{(x,y)\to(0,0)} \sin x \ln y$

5. Consider the function $f(x, y) = \frac{x^2+y^2}{x^2+y^4}$. Show that $\lim_{(x,y)\to(0,0)} f(x, y)$ does not exist.

6. Let $f(x, y) = \begin{cases} x^2 - y^2, & \text{if } x+y<0 \\ 2\,x + y, & \text{if } x+y \geq 0 \end{cases}$.

Is f continuous?

7. Let $f(x, y) = \frac{x\,y}{x^2+y^2}$. The domain of f is the whole plane without the origin. Is it possible to define $f(0, 0)$ so that f is continuous everywhere? Plot the graph of f to support your conclusions.

8. The domain of $f(x, y) = \frac{x\,y}{x+y}$ is the whole plane without the line $y = -x$. Is it possible to define $f(0, 0)$ so that f is continuous everywhere? Plot the graph of f to support your conclusions.

■ 14.3 Partial Derivatives

Students should read Section 14.3 of Rogawski's *Calculus* [1] for a detailed discussion of the material presented in this section.

Recall that the *Mathematica* command for the partial derivative of a function **f** with respect to **x** is **D[f, x]**, and **D[f,{x,n}]** gives the **n**th partial derivative of **f** with respect to **x**. The multiple (mixed) partial derivative of **f** with respect to $\mathbf{x}_1$, $\mathbf{x}_2$, $\mathbf{x}_3$, ... is obtained by **D[f,** $\mathbf{x}_1$, $\mathbf{x}_2$, $\mathbf{x}_3$, **...]**. We can access this command from **BasicMathInput**. The symbols are $\partial_{\square}\,\square$ and $\partial_{\square,\square}\,\square$.

Example 14.20. Find the first partial derivatives of $x^3 + y^2$ with respect to x and y .

Solution: We give two methods of input.

Method 1: We can type all the inputs and the command as follows:

```
In[262]:= Clear[x, y]
          D[x^3 + y^2, x]
Out[263]= 3 x^2
In[264]:= D[x^3 + y^2, y]
Out[264]= 2 y
```

Method 2: We can use the **BasicInput** palette to enter the inputs.

In[265]:= $\partial_x \left(x^3 + y^2\right)$

Out[265]= $3\,x^2$

In[266]:= $\partial_y \left(x^3 + y^2\right)$

Out[266]= $2\,y$

Example 14.21. Find the four second partial derivatives of $x^3 \sin(y) + e^{x\,y}$.

Solution: Let $z = x^3 \sin y + e^{x\,y}$. We again demonstrate two methods of input.

Method 1:

We can find z_{xx} by

In[267]:=
```
Clear[x, y]
D[x^3 * Sin[y] + E^(x * y), {x, 2}]
```

Out[268]= $e^{x y} y^2 + 6 x \text{Sin}[y]$

We can find z_{yy} by

In[269]:=
```
D[x^3 * Sin[y] + E^(x * y), {y, 2}]
```

Out[269]= $e^{x y} x^2 - x^3 \text{Sin}[y]$

We can find z_{xy} by

In[270]:=
```
D[x^3 * Sin[y] + E^(x * y), x, y]
```

Out[270]= $e^{x y} + e^{x y} x y + 3 x^2 \text{Cos}[y]$

z_{yx} is given by

In[271]:=
```
D[x^3 * Sin[y] + E^(x * y), y, x]
```

Out[271]= $e^{x y} + e^{x y} x y + 3 x^2 \text{Cos}[y]$

NOTE: Clairaut's Theorem states that if the mixed partial derivatives $f_{x y}$ and $f_{y x}$ are continuous at a point (x, y), then they are equal: $f_{x y} = f_{y x}$. The last two outputs confirm Clairaut's Theorem for this particular example.

Method 2: Here is the input using the palette symbol $\partial_{\square,\square}\square$:

In[272]:=

$\textbf{Clear}[x, y]$

$\partial_{x,x}\left(x^3 * \textbf{Sin}[y] + e^{x*y}\right)$

$\partial_{y,y}\left(x^3 * \textbf{Sin}[y] + e^{x*y}\right)$

$\partial_{x,y}\left(x^3 * \textbf{Sin}[y] + e^{x*y}\right)$

$\partial_{y,x}\left(x^3 * \textbf{Sin}[y] + e^{x*y}\right)$

Out[273]= $e^{x y} y^2 + 6 x \text{Sin}[y]$

Out[274]= $e^{x y} x^2 - x^3 \text{Sin}[y]$

Out[275]= $e^{x y} + e^{x y} x y + 3 x^2 \text{Cos}[y]$

Out[276]= $e^{x y} + e^{x y} x y + 3 x^2 \text{Cos}[y]$

Example 14.22. Evaluate the first partial derivatives of $x y + y z^2 + x z$ at $(-1, 2, 3)$.

Solution: Recall that **Expr /.** $\{x_1 \to a_1, x_2 \to a_2, x_3 \to a_3, \ldots\}$ is the command for substituting x_1 by a_1, x_2 by a_2, x_3 by a_3, ... , in **Expr**.

In[277]:=
```
Clear[x,y,z]
D[x*y + y*z^2 + x*z,x]/.{x-> -1, y->2, z->3}
```

Out[278]= 5

In[279]:=
```
D[x * y + y * z^2 + x * z, y] /. {x -> -1, y -> 2, z -> 3}
```

Out[279]= 8

In[280]:=
```
D[x * y + y * z^2 + x * z, z] /. {x -> -1, y -> 2, z -> 3}
```
Out[280]= 11

Example 14.23. Let $f(x, y, z) = y e^x + x e^{-y} \ln z$. Find f_{xxx}, f_{xyz}, f_{xzz}, f_{zxz}, and f_{zzx}.

Solution: First, we define $f(x, y, z)$ in *Mathematica*. We can use the $\partial_{\square,\square}\square$ notation. Since the palette gives only two boxes for the variables, we need to add one more box. This can be done by using **CTRL +, (comma)**, that is, hold the CONTROL key and press the COMMA button. Note also that the command **D[f[x,y,z],x,y,z]** gives f_{xyz}. We demonstrate both methods.

In[281]:=
```
Clear[x, y, z, f]
f[x_, y_, z_] := y * e^x + x * Log[z] * e^-y
```

In[283]:= $\partial_{x,x,x}$ f[x, y, z]

Out[283]= $e^x y$

In[284]:= $\partial_{x,y,z}$ f[x, y, z]

Out[284]= $-\frac{e^{-y}}{z}$

In[285]:= $\partial_{x,z,z}$ f[x, y, z]

Out[285]= $-\frac{e^{-y}}{z^2}$

In[286]:=
```
D[f[x, y, z], z, x, z]
```
Out[286]= $-\frac{e^{-y}}{z^2}$

In[287]:=
```
D[f[x, y, z], z, z, x]
```
Out[287]= $-\frac{e^{-y}}{z^2}$

Example 14.24. Let $f(x, y) = x y \frac{x^2-y^2}{x^2+y^2}$ if $(x, y) \neq (0, 0)$ and $f(0, 0) = 0$.

a) Find $f_x(x, y)$ and $f_y(x, y)$ for $(x, y) \neq (0, 0)$.

b) Use the limit definition to find $f_x(0, 0)$ and $f_y(0, 0)$.

c) Find $f_{xy}(x, y)$ and $f_{yx}(x, y)$ for $(x, y) \neq (0, 0)$.

d) Use the limit definition to find $f_{xy}(0, 0)$ and $f_{yx}(0, 0)$.

Solution: We will first define f using the **If** command.

In[288]:=
```
Clear[x, y, f, fx, fy, fxy, fyx]
```
$$f[x\_, y\_] = \text{If}\left[\{x, y\} \neq \{0, 0\}, \; x\, y \frac{x^2 - y^2}{x^2 + y^2}, 0\right]$$

Out[289]= $\text{If}\left[\{x, y\} \neq \{0, 0\}, \frac{x\, y\, (x^2 - y^2)}{x^2 + y^2}, 0\right]$

a) Let **fx** and **fy** denote the partial derivatives with respect to x and y, respectively. Then

In[290]:=
```
fx[x_, y_] = D[f[x, y], x]
fy[x_, y_] = D[f[x, y], y]
```

Out[290]= $$\text{If}\left[\{x, y\} \neq \{0, 0\}, \left(-\frac{2\,x\,(x^2-y^2)}{(x^2+y^2)^2}+\frac{2\,x}{x^2+y^2}\right)(x\,y)+\frac{y\,(x^2-y^2)}{x^2+y^2}, 0\right]$$

Out[291]= $$\text{If}\left[\{x, y\} \neq \{0, 0\}, \left(-\frac{2\,y\,(x^2-y^2)}{(x^2+y^2)^2}-\frac{2\,y}{x^2+y^2}\right)(x\,y)+\frac{x\,(x^2-y^2)}{x^2+y^2}, 0\right]$$

If we use the **FullSimplify** command to simplify the preceding output, we get

In[292]:=
```
FullSimplify[fx[x, y]]
FullSimplify[fy[x, y]]
```

Out[292]= $$\begin{cases} \frac{y\,(x^4+4\,x^2\,y^2-y^4)}{(x^2+y^2)^2} & x \neq 0 \,||\, y \neq 0 \\ 0 & \text{True} \end{cases}$$

Out[293]= $$\begin{cases} \frac{x\,(x^4-4\,x^2\,y^2-y^4)}{(x^2+y^2)^2} & x \neq 0 \,||\, y \neq 0 \\ 0 & \text{True} \end{cases}$$

Thus, $f_x(x, y) = \frac{y(x^4+4\,x^3\,y^2-y^4)}{(x^2+y^2)^2}$ and $f_y(x, y) = \frac{x(x^4-4\,x^2\,y^2-y^4)}{(x^2+y^2)^2}$ if $(x, y) \neq (0, 0)$.

b) We use the limit definition $f_x(0, 0) = \lim_{h\to 0} \frac{f(0+h,0)-f(0,0)}{h}$ and $f_y(0, 0) = \lim_{k\to 0} \frac{f(0,0+k)-f(0,0)}{k}$ to find the partial derivatives at (0, 0).

In[294]:=
```
Clear[h, k]
Limit[(f[0 + h, 0] - f[0, 0])/h, h → 0]
```

Out[295]= 0

In[296]:=
```
Limit[(f[0, 0 + k] - f[0, 0])/k, k → 0]
```

Out[296]= 0

Hence, $f_x(0, 0) = 0$ and $f_y(0, 0) = 0$.

c) To find the mixed second partial derivatives, we use **fx** and **fy** from the outputs in part a). Note that the **FullSimplify** command is used to to get a simplified form of the mixed partial derivatives.

In[297]:=
```
fxy[x_, y_] = FullSimplify[D[fx[x, y], y]]
fyx[x_, y_] = FullSimplify[D[fy[x, y], x]]
```

Out[297]= $$\begin{cases} \frac{(x-y)\,(x+y)\,(x^4+10\,x^2\,y^2+y^4)}{(x^2+y^2)^3} & x \neq 0 \,||\, y \neq 0 \\ 0 & \text{True} \end{cases}$$

Out[298]= $$\begin{cases} \frac{(x-y)\,(x+y)\,(x^4+10\,x^2\,y^2+y^4)}{(x^2+y^2)^3} & x \neq 0 \,||\, y \neq 0 \\ 0 & \text{True} \end{cases}$$

Thus, $f_{x\,y} = \frac{(x-y)(x+y)\left(x^4+10\,x^2\,y^2+y^4\right)}{\left(x^2+y^2\right)^3}$ and $f_{y\,x} = \frac{(x-y)(x+y)\left(x^4+10\,x^2\,y^2+y^4\right)}{\left(x^2+y^2\right)^3}$ for $(x,\ y) \neq (0,\ 0)$. Note that these two functions are equal for $(x,\ y) \neq (0,\ 0)$ in conformity with Clairaut's Theorem, since both are continuous when $(x,\ y) \neq (0,\ 0)$.

d) We use the limit definition of a partial derivative to compute $f_{xy}(0,\ 0)$ and $f_{yx}(0,\ 0)$. Recall that we have defined f_x as **fx[x,y]** and f_y as **fy[x,y]**.

Then $f_{xy}(0,\ 0)$ is given by

In[299]:= **Limit[(fx[0 , 0 + k] - fx[0, 0])/k, k → 0]**

Out[299]= -1

and $f_{yx}(0,\ 0)$ is given by

In[300]:= **Limit[(fy[0 + h, 0] - fy[0, 0])/h, h → 0]**

Out[300]= 1

Thus, $f_{xy}(0,\ 0) = -1$and $f_{yx}(0,\ 0) = 1$. Note that this implies that the mixed partial derivatives are not continuous at $(x,\ y) = (0,\ 0)$. To see this graphically, first consider the following graph of f, which confirms that f has partial derivatives everywhere.

In[301]:= **Plot3D[f[x, y], {x, -3, 3}, {y, -3, 3}, ImageSize → {250}]**

Out[301]=



Here are the graphs of f_x and f_y, which now show why the second mixed partials at the origin are not equal.

In[302]:=
```
Clear[plot1, plot2]
plot1 = Plot3D[fx[x, y], {x, -3, 3}, {y, -3, 3},
   PlotStyle → Red, AxesLabel → {"Graph of z=fx", None, None}] ;
plot2 = Plot3D[fy[x, y], {x, -3, 3}, {y, -3, 3}, PlotStyle → Blue,
   AxesLabel → {"Graph of z=fy", None, None}] ;
Show[GraphicsArray[{plot1, plot2}], ImageSize → {420}]
```

GraphicsArray::obs : GraphicsArray is obsolete. Switching to GraphicsGrid. ≫

Out[305]=



In addition, the graphs of f_{xy} and f_{yx} show the mixed partials are not continuous at the origin. This is the main reason why the inequalities of the mixed partials at the origin does not contradict Clairaut's Theorem.

In[306]:=
```
Clear[plot1, plot2]
plot1 = Plot3D[fxy[x, y], {x, -3, 3}, {y, -3, 3},
   PlotStyle → Red, AxesLabel → {"Graph of z=fxy", None, None}];
plot2 = Plot3D[fyx[x, y], {x, -3, 3}, {y, -3, 3}, PlotStyle → Blue,
   AxesLabel → {"Graph of z=fyx", None, None}];
Show[GraphicsArray[{plot1, plot2}], ImageSize → {420}]
```

GraphicsArray::obs : GraphicsArray is obsolete. Switching to GraphicsGrid. ≫

Out[309]=



Exercises

1. Let $f(x, y) = \frac{(x-y)^2}{x^2+y^2}$. Find:

a. $f_x(1,0)$ b. $f_y(1, 0)$ c. f_{xy} d. f_{yx} e. f_{xxy}

2. Find the first partial derivatives of $z = x^3 y^2$ with respect to x and y.

3. Find the four second partial derivatives of $x^2 \cos(y) + \tan(x e^y)$.

4. Evaluate the first partial derivatives of $f(x, y, z) = e^{-z} xy + yz^2 + xz$ at $(-1, 2, 3)$.

5. Let $f(x, y, z) = \frac{x^4 y^3}{z^2+\sin x}$. Find f_{xxx}, f_{xyz}, f_{xzz}, f_{zxz}, and f_{zzx}.

6. Let $f(x, y) = \frac{x y^2}{x^2+y^4}$ if $(x, y) \neq (0, 0)$ and $f(0, 0) = 0$.

a. Find $f_x(x, y)$ and $f_y(x, y)$ for $(x, y) \neq (0, 0)$.

b. Use the limit definition to find $f_x(0, 0)$ and $f_y(0, 0)$.

c. Find $f_{xy}(x, y)$ and $f_{yx}(x, y)$ for $(x, y) \neq (0, 0)$.

d. Use the limit definition to find $f_{xy}(0, 0)$ and $f_{yx}(0, 0)$.

14.4 Tangent Planes

Students should read Section 14.4 of Rogawski's *Calculus* [1] for a detailed discussion of the material presented in this section.

Let $z = f(x, y)$ be a function of two variables. The equation of the *tangent plane* at the point $(a, b, f(a, b))$ is given by

$$z = f_x(a, b)(x-a) + f_y(a, b)(y-b) + f(a, b)$$

Example 14.25. Let $f(x, y) = x^2 + y^2$.
a) Find the equation of the tangent plane to the graph of f at the point $(2, 1, 3)$.
b) Plot the graph of f and its tangent plane at $(2, 1, 3)$.

Solution: Here, $a = 2$, $b = 1$.
a)

```
In[310]:= Clear[f, x, y, z]
          f[x_, y_] = x^2 + y^2
Out[311]= x^2 + y^2
```

Thus, the equation the of the tangent plane is

```
In[312]:= A = ∂x f[x, y] /. {x → 2, y → 1};
          B = ∂y f[x, y] /. {x → 2, y → 1};
          z = A (x - 2) + B (y - 1) + f[2, 1];
          Simplify[z]
Out[315]= -5 + 4 x + 2 y
```

b) Here is a plot of the graph of f:

```
In[316]:= plot1 = Plot3D[{f[x, y], z}, {x, -10, 10}, {y, -10, 10}, PlotStyle → {Blue, Green}];
          plot2 = ListPointPlot3D[{ {2, 1, 3}}, PlotStyle → {Red, PointSize[Large]} ];
          Show[plot1, plot2, ImageSize → {250}, ImagePadding → {{15, 15}, {15, 15}}]
```

Out[318]=



Example 14.26. Let $f(x, y) = x^2 y - 6 x y^2 + 3 y$. Find the points where the tangent plane to the graph of f is parallel to the xy-plane.

Solution: For the tangent plane to be parallel to the xy-plane, we must have $f_x = 0$ and $f_y = 0$.

```
In[319]:= Clear[f, x, y ]
          f[x_, y_] = x^2 y - 6 x y^2 + 3 y
```

Out[320]= $3 y + x^2 y - 6 x y^2$

A tangent plane is parallel to the xy-plane at

```
In[321]:= Solve[{ D[f[x, y], x] == 0, D[f[x, y], y] == 0}]
```

Out[321]= $\left\{\left\{y \to -\frac{1}{3}, x \to -1\right\}, \left\{y \to 0, x \to -i\sqrt{3}\right\}, \left\{y \to 0, x \to i\sqrt{3}\right\}, \left\{y \to \frac{1}{3}, x \to 1\right\}\right\}$

Rotate the following graph to see the points of tangencies.

In[322]:=
```
Plot3D[{f[x, y], f[-1, -1 / 3], f[1, 1 / 3]}, {x, -1, 1},
 {y, -1, 1}, PlotStyle → {LightBlue, Green, Red}, PlotRange → All,
 ImageSize → {250}, ImagePadding → {{15, 15}, {15, 15}}]
```

Out[322]=



Exercises

1. Let $f(x, y) = x^3 y + x y^2 - 3x + 4$.
a) Find a set of parametric equations of the normal line and an equation of the tangent plane to the surface at the point (1, 2).
b) Graph the surface, the normal line, and the tangent plane found in a).

2. Let $f(x, y) = x^2 + y^2$.
a. Find the equation of the tangent plane to the graph of f at the point (2, 1, 5).
b. Plot the graph of f and its tangent plane at (2, 1, 5).

3. Let $f(x, y) = e^{-y/x}$.
a. Find the equation of the tangent plane to the graph of f at the point (1, 0, 1).
b. Plot the graph of f and its tangent plane at (1, 0, 1).

4. Let $f(x, y) = \cos(x y)$. Find the points where the tangent plane to the graph of f is parallel to the xy-plane.

14.5 Gradient and Directional Derivatives

Students should read Section 14.5 of Rogawski's *Calculus* [1] for a detailed discussion of the material presented in this section.

Recall that the notation for a vector such as $\mathbf{u} = 2\mathbf{i} + 5\mathbf{j} - 6\mathbf{k}$ in *Mathematica* is {2,5,-6}. The command for the *dot product* of two vectors **u** and **v** is obtained by typing **u.v**.

The *gradient* of f, denoted by ∇f, at (a, b) can be obtained by evaluating $\nabla f(a, b) = \langle \partial_x f(a, b),\ \partial_y f(a, b)\rangle$.

The *directional derivative* of f at (a, b) *in the direction of a unit vector* **u** is given by $D_{\mathbf{u}} f = \nabla f(a, b) \cdot \mathbf{u}$.

Example 14.27. Find the gradient and directional derivative of $f(x, y) = x^2 \sin 2y$ at the point $\left(1, \frac{\pi}{2}, 0\right)$ in the direction of $\mathbf{v} = \left\langle \frac{3}{5}, -\frac{4}{5} \right\rangle$.

Solution:

In[323]:=
```
Clear[f, v]
f[x_, y_] := x^2 * Sin[2 y]
v = {3/5, -4/5}
```

Out[325]= $\left\{\frac{3}{5}, -\frac{4}{5}\right\}$

The gradient of f at $\left(1, \frac{\pi}{2}\right)$ is

In[326]:=
```
∇f = {∂x f[x, y], ∂y f[x, y]} /. {x -> 1, y -> π/2}
```

Out[326]= $\{0, -2\}$

Since **v** is a unit vector, the directional derivative is given by

In[327]:=
```
direcderiv = v.∇f
```

Out[327]= $\frac{8}{5}$

Example 14.28. Find the gradient and directional derivative of $f(x, y, z) = x\,y + y\,z + x\,z$ at the point $(1, 1, 1)$ in the direction of $\mathbf{v} = 2\,\mathbf{i} + \mathbf{j} - \mathbf{k}$.

Solution:

In[328]:=
```
Clear[x, y, z]
w = x * y + y * z + x * z
v = {2, 1, -1}
```

Out[329]= $x\,y + x\,z + y\,z$

Out[330]= $\{2, 1, -1\}$

We normalize **v**:

In[331]:=
```
unitvector = v / Norm[v]
```

Out[331]= $\left\{\sqrt{\frac{2}{3}}, \frac{1}{\sqrt{6}}, -\frac{1}{\sqrt{6}}\right\}$

The gradient of $w = f(x, y, z)$ at $(1, 1, 1)$ is

In[332]:=
```
∇w = {D[w, x], D[w, y], D[w, z]} /. {x -> 1, y -> 1, z -> 1}
```

Out[332]= $\{2, 2, 2\}$

Hence, the directional derivative is given by

In[333]:=
```
direcderiv = unitvector.∇w
```

Out[333]= $2\sqrt{\frac{2}{3}}$

Example 14.29. Plot the gradient vector field and the level curves of the function $f(x, y) = x^2 \sin 2\,y$.

Solution:

In[334]:=
```
Clear[f, fx, fy, x, y]
f[x_, y_] = x^2 - 3 x y + y - y^2
fx = D[f[x, y], x]
fy = D[f[x, y], y]
```

Out[335]= $x^2 + y - 3\,x\,y - y^2$

Out[336]= $2\,x - 3\,y$

Out[337]= $1 - 3\,x - 2\,y$

Thus, the gradient vector field is $\nabla f(x, y) = \langle 2\,x - 3\,y, 1 - 3\,x - 2\,y \rangle$. To plot this vector field, we need to download the package **VectorFieldPlots**, which is done by evaluating

In[338]:=
```
Needs["VectorFieldPlots`"]
```

General::obspkg :
VectorFieldPlots` is now obsolete. The legacy version being loaded may conflict with current Mathematica functionality. See the Compatibility Guide for updating information. ≫

Here is a plot of some level curves and the gradient field.

In[339]:=
```
Clear[plot1, plot2]
plot1 = ContourPlot[f[x, y], {x, -5, 5}, {y, -4, 4},
   Axes → True, Frame → False, Contours → 15, ColorFunction → Hue] ;
plot2 = VectorFieldPlot[{fx, fy}, {x, -5, 5}, {y, -4, 4}, Axes → True, Frame → False];
Show[plot1, plot2, ImageSize → {250}]
```

Out[342]=



Example 14.30. Let the temperature T at a point (x, y) on a metal plate be given by $T(x, y) = \frac{x}{x^2+y^2}$.

a) Plot the graph of the temperature.
b) Find the rate of change of temperature at (3, 4), in the direction of $\mathbf{v} = \mathbf{i} - 2\,\mathbf{j}$.
c) Find the unit vector in the direction of which the temperature increases most rapidly at (3, 4).
d) Find the maximum rate of increase in the temperature at (3, 4).

Solution:

a) Here is the graph of T.

In[343]:=
```
T[x_, y_] = x/(x^2 + y^2)
```

Out[343]= $\dfrac{x}{x^2+y^2}$

In[344]:=
```
graphofT =
 Plot3D[T[x, y], {x, -5, 5}, {y, -5, 5}, BoxRatios → {1, 1, 1}, ImageSize → Small]
```

Out[344]=



b) Let $\mathbf{u} = \frac{\mathbf{v}}{\|\mathbf{v}\|}$. Then $\mathbf{u}$ is a unit vector and the rate of change in temperature at (3, 4) in the direction of $\mathbf{v}$ is given by $D_{\mathbf{u}}\, T(3, 4) = \nabla f(3, 4) \cdot \mathbf{u}$.

In[345]:=
```
∇T = {D[T[x, y], x], D[T[x, y], y]}
v = {1, -2}
u = v/Sqrt[v.v]
u.∇T /. {x -> 3, y -> 4} // N
```

Out[345]= $\left\{-\dfrac{2x^2}{(x^2+y^2)^2}+\dfrac{1}{x^2+y^2},\ -\dfrac{2xy}{(x^2+y^2)^2}\right\}$

Out[346]= $\{1, -2\}$

Out[347]= $\left\{\dfrac{1}{\sqrt{5}},\ -\dfrac{2}{\sqrt{5}}\right\}$

Out[348]= 0.0393548

Thus, the rate of change at (3, 4) in the direction $\mathbf{v}$ is 0.0393548. NOTE: The command **//N** in the last line of the previous input converts the output to decimal form.

c) The unit vector in the direction of which the temperature increases most rapidly at (3, 4) is given by

In[349]:=
```
∇T/Norm[∇T] /. {x -> 3, y -> 4}
```
Out[349]= $\left\{\frac{7}{25}, -\frac{24}{25}\right\}$

d) The maximum rate of increase in the temperature at (3,4) is the norm of the gradient at this point. This can be obtained by:

In[350]:=
```
Norm[∇T] /. {x -> 3, y -> 4}
```
Out[350]= $\frac{1}{25}$

Exercises

1. Find the gradient and directional derivative of $f(x, y) = \sin^{-1}(x\,y)$ at the point $\left(1, 1, \frac{\pi}{2}\right)$ in the direction of $\mathbf{v} = \langle 1, -1\rangle$.

2. Let $T(x, y) = e^{x\,y - y^2}$.
a. Find $\nabla T(x, y)$.
b. Find the directional derivative of $T(x, y)$ at the point (3, 5) in the dierection of $\mathbf{u} = 1/2\,\mathbf{i} + \sqrt{3}/2\,\mathbf{j}$.
c. Find the direction of greatest increase in T from the point (3, 5).

3. Plot the gradient vector field and the level curves of the function a $f(x, y) = \cos x \sin^2 y$.

4. Find the gradient and directional derivative of $f(x, y, z) = x\,y\,e^{y\,z} + \sin(x\,z)$ at the point (1, 1, 0) in the direction of $\mathbf{v} = \mathbf{i} - \mathbf{j} - \mathbf{k}$.

14.6 The Chain Rule

Students should read Section 14.6 of Rogawski's *Calculus* [1] for a detailed discussion of the material presented in this section.

Example 14.31. Let $x = t^2 + s$, $y = t + s^2$ and $z = x \sin y$. Find the first partial derivatives of z with respect to s and t.

Solution:

In[351]:=
```
Clear[x, y, z, s, t]
x = t^2 + s
y = t + s^2
z = x Sin[y]
```
Out[352]= $s + t^2$

Out[353]= $s^2 + t$

Out[354]= $(s + t^2)\,\mathrm{Sin}[s^2 + t]$

In[355]:=
```
D[z, s]
```
Out[355]= $2\,s\,(s + t^2)\,\mathrm{Cos}[s^2 + t] + \mathrm{Sin}[s^2 + t]$

In[356]:=
```
D[z, t]
```

Out[356]= $(s+t^2)\,\mathrm{Cos}[s^2+t]+2\,t\,\mathrm{Sin}[s^2+t]$

Example 14.32. Find the partial derivatives of z with respect to x and y assuming that the equation $x^2 z - y z^2 = x y$ defines z as a function of x and y.

Solution:

In[357]:=
```
Clear[x, y, z, r, t, s]
eq = x^2 z[x, y] - y z[x, y]^2 == x y

Solve[D[eq, x], D[z[x, y], x]]
Solve[D[eq, y], D[z[x, y], y]]
```

Out[358]= $x^2\,z[x,y] - y\,z[x,y]^2 == x\,y$

Out[359]= $\left\{\left\{z^{(1,0)}[x,y] \to \frac{-y+2\,x\,z[x,y]}{-x^2+2\,y\,z[x,y]}\right\}\right\}$

Out[360]= $\left\{\left\{z^{(0,1)}[x,y] \to \frac{x+z[x,y]^2}{x^2-2\,y\,z[x,y]}\right\}\right\}$

Example 14.33. Let $f(x, y, z) = F(r)$, where $r = \sqrt{x^2+y^2+z^2}$ and F is a twice differentiable function of one variable.

a) Show that $\nabla f = F'(r)\frac{1}{r}(x\,\mathbf{i} + y\,\mathbf{j} + z\,\mathbf{k})$.

b) Find the Laplacian of f.

Solution:

a)

In[361]:=
```
Clear[x, y, z, r, f, F]
f[x_, y_, z_] = F[r]

r = Sqrt[x^2 + y^2 + z^2]
```

Out[362]= F[r]

Out[363]= $\sqrt{x^2+y^2+z^2}$

Here is the gradient of f:

In[364]:=
```
gradf = {D[f[x, y, z], x], D[f[x, y, z], y], D[f[x, y, z], z]}
```

Out[364]= $\left\{\frac{x\,F'\left[\sqrt{x^2+y^2+z^2}\right]}{\sqrt{x^2+y^2+z^2}}, \frac{y\,F'\left[\sqrt{x^2+y^2+z^2}\right]}{\sqrt{x^2+y^2+z^2}}, \frac{z\,F'\left[\sqrt{x^2+y^2+z^2}\right]}{\sqrt{x^2+y^2+z^2}}\right\}$

With $r = \sqrt{x^2+y^2+z^2}$, the preceding output becomes

$$\nabla f(x, y, z) = \left\langle \frac{x\,F'(r)}{r}, \frac{y\,F'(r)}{r}, \frac{z\,F'(r)}{r} \right\rangle = F'(r)\frac{1}{r}\langle x, y, z\rangle$$

which proves part a).

b) Recall that the Laplacian of f, denoted by Δf, is defined by $\Delta f = f_{xx} + f_{yy} + f_{zz}$.

In[365]:=
```
D[f[x, y, z], {x, 2}] + D[f[x, y, z], {y, 2}] + D[f[x, y, z], {z, 2}]
```

Out[365]=
$$-\frac{x^2 F'\left[\sqrt{x^2+y^2+z^2}\right]}{\left(x^2+y^2+z^2\right)^{3/2}}-\frac{y^2 F'\left[\sqrt{x^2+y^2+z^2}\right]}{\left(x^2+y^2+z^2\right)^{3/2}}-\frac{z^2 F'\left[\sqrt{x^2+y^2+z^2}\right]}{\left(x^2+y^2+z^2\right)^{3/2}}+\frac{3 F'\left[\sqrt{x^2+y^2+z^2}\right]}{\sqrt{x^2+y^2+z^2}}+\frac{x^2 F''\left[\sqrt{x^2+y^2+z^2}\right]}{x^2+y^2+z^2}+\frac{y^2 F''\left[\sqrt{x^2+y^2+z^2}\right]}{x^2+y^2+z^2}+\frac{z^2 F''\left[\sqrt{x^2+y^2+z^2}\right]}{x^2+y^2+z^2}$$

We simplify this to get

In[366]:=
```
Simplify[%]
```

Out[366]=
$$\frac{2 F'\left[\sqrt{x^2+y^2+z^2}\right]}{\sqrt{x^2+y^2+z^2}}+F''\left[\sqrt{x^2+y^2+z^2}\right]$$

which is the same as $\frac{2}{r} F'[r] + F''[r]$.

Exercises

1. Let $x = u^2 + \sin v$, $y = u\, e^{v/u}$, and $z = y^3 \ln x$. Find the first partial derivatives of z with respect to u and v.

2. Find the partial derivatives of z with respect to x and y assuming that the equation $x^2 z - y z^2 = x y$ defines z as a function of x and y.

3. Find an equation of the tangent plane to the surface $x z + 2 x^2 y + y^2 z^3 = 11$ at (2, 1, 1).

14.7 Optimization

Students should read Section 14.7 of Rogawski's *Calculus* [1] for a detailed discussion of the material presented in this section.

Second Derivative Test: Suppose $f_x(a, b) = 0$ and $f_y(a, b) = 0$. Define

$$D(x, y) = f_{xx} f_{yy} - \left(f_{xy}\right)^2$$

The function D is called the *discriminant function.*

i) If $D(a, , b) > 0$ and $f_{xx}(a, b) > 0$, then $f(a, b)$ is a local minimum value.
ii) If $D(a, , b) > 0$ and $f_{xx}(a, b) < 0$, then $f(a, b)$ is a local maximum value.
iii) If $D(a, , b) < 0$, then $(a, b, f(a, b))$ is a saddle point on the graph of f.
iv) If $D(a, b) = 0$, then no conclusion can be drawn about the the point (a, b).

Example 14.34. Let $f(x, y) = x^4 - 4 x y + 2 y^2$.
a) Find all critical points of f.
b) Use the second derivative test to classify the critical points as local minimum, local maximum, saddle point, or neither.

Solution: Since **D** is used in *Mathematica* as the command for derivative, we will use **disc** for the discriminant function D.

In[367]:=
```
Clear[f, x, y]
f[x_, y_] = x^4 - 4 x y + 2 y^2
```

Out[368]= $x^4 - 4 x y + 2 y^2$

a) The critical points are given by

In[369]:=
```
cp = Solve[{D[f[x, y], x] == 0, D[f[x, y], y] == 0}]
```

Out[369]= {{y → -1, x → -1}, {y → 0, x → 0}, {y → 1, x → 1}}

b)

In[370]:=
```
Clear[fxx, disc]
fxx[x_, y_] = D[f[x, y], {x, 2}]
disc[x_, y_] = D[f[x, y], {x, 2}] * D[f[x, y], {y, 2}] - (D[D[f[x, y], x], y])^2
```

Out[371]= $12 x^2$

Out[372]= $-16 + 48 x^2$

In[373]:=
```
TableForm[Table[{ cp[[k, 2, 2]],  cp[[k, 1, 2]] ,
   disc[cp[[k, 2, 2]], cp[[k, 1, 2]]], fxx[cp[[k, 2, 2]], cp[[k, 1, 2]]],
   f[cp[[k, 2, 2]], cp[[k, 1, 2]]]}, {k, 1, Length[cp]}],
 TableHeadings → {{}, {"x ", "y  ", "  D(x,y) ", "  fxx  ", "f(x,y)"}}]
```

Out[373]//TableForm=

x	y	D(x,y)	f_{xx}	f(x,y)
-1	-1	32	12	-1
0	0	-16	0	0
1	1	32	12	-1

By the second derivative test, we conclude that f has a local minimum value of -1 at $(-1, -1)$ and $(1, 1)$, and a saddle point at $(0, 0)$.

Here is the graph of f and the relevant points.

In[374]:=
```
Clear[plot1, plot2]
plot1 =
  Plot3D[f[x, y], {x, -2, 2}, {y, -2, 2}, PlotStyle → LightBlue, PlotRange → 10];
plot2 = Graphics3D[{PointSize[Large], Red,
    Point[Table[{ cp[[k, 2, 2]],  cp[[k, 1, 2]] ,  f[cp[[k, 2, 2]], cp[[k, 1, 2]]]},
      {k, 1, Length[cp]}]]}, PlotRange → 10];
Show[plot1, plot2, ImageSize → {250}]
```

Out[377]=



Example 14.35. Let $f(x, y) = x^3 + y^4 - 6x - 2y^2$.
a) Find all critical points of f.
b) Use the second derivative test to classify the critical points as local minimum, local maximum, saddle point, or neither.

Solution: Again, we will use **disc** to denote the discriminant function D since the letter **D** is used in *Mathematica* for the derivative command.

In[378]:=
```
Clear[f, x, y]
f[x_, y_] = x^3 + y^4 - 6 x - 2 y^2
```

Out[379]= $-6x + x^3 - 2y^2 + y^4$

a) The critical points are given by

In[380]:=
```
cp = Solve[{D[f[x, y], x] == 0, D[f[x, y], y] == 0}]
```

Out[380]= $\{\{y \to -1, x \to -\sqrt{2}\}, \{y \to -1, x \to \sqrt{2}\}, \{y \to 0, x \to -\sqrt{2}\}, \{y \to 0, x \to \sqrt{2}\}, \{y \to 1, x \to -\sqrt{2}\}, \{y \to 1, x \to \sqrt{2}\}\}$

b)

In[381]:=
```
Clear[fxx, disc]
fxx[x_, y_] = D[f[x, y], {x, 2}]
disc[x_, y_] = D[f[x, y], {x, 2}] * D[f[x, y], {y, 2}] - (D[D[f[x, y], x], y])^2
```

Out[382]= $6x$

Out[383]= $6x(-4 + 12y^2)$

In[384]:=
```
TableForm[Table[{ cp[[k, 2, 2]],  cp[[k, 1, 2]] ,
   disc[cp[[k, 2, 2]], cp[[k, 1, 2]]], fxx[cp[[k, 2, 2]], cp[[k, 1, 2]]],
   f[cp[[k, 2, 2]], cp[[k, 1, 2]]]}, {k, 1, Length[cp]}],
  TableHeadings → {{}, {"x ", "y  ", "  D(x,y) ", "  fxx  ", "f(x,y)"}}]
```

Out[384]//TableForm=

x	y	D(x,y)	f_{xx}	f(x,y)
$-\sqrt{2}$	-1	$-48\sqrt{2}$	$-6\sqrt{2}$	$-1+4\sqrt{2}$
$\sqrt{2}$	-1	$48\sqrt{2}$	$6\sqrt{2}$	$-1-4\sqrt{2}$
$-\sqrt{2}$	0	$24\sqrt{2}$	$-6\sqrt{2}$	$4\sqrt{2}$
$\sqrt{2}$	0	$-24\sqrt{2}$	$6\sqrt{2}$	$-4\sqrt{2}$
$-\sqrt{2}$	1	$-48\sqrt{2}$	$-6\sqrt{2}$	$-1+4\sqrt{2}$
$\sqrt{2}$	1	$48\sqrt{2}$	$6\sqrt{2}$	$-1-4\sqrt{2}$

By the second derivative test we conclude that f has local maximum value of $4\sqrt{2}$ at $\left(-\sqrt{2}, 0\right)$, local minimum value of $-1-4\sqrt{2}$ at $\left(\sqrt{2}, -1\right)$ and $\left(\sqrt{2}, 1\right)$, and saddle points at $\left(-\sqrt{2}, -1\right)$, $\left(\sqrt{2}, 0\right)$, and $\left(-\sqrt{2}, 1\right)$.

Here is the graph of f and the relevant points.

In[385]:=
```
Clear[plot1, plot2]
plot1 = Plot3D[f[x, y], {x, -2.5, 2.5},
   {y, -2.5, 2.5}, PlotStyle → LightBlue, PlotRange → 10];
plot2 = Graphics3D[{PointSize[Large], Red,
    Point[Table[{ cp[[k, 2, 2]],  cp[[k, 1, 2]] ,  f[cp[[k, 2, 2]], cp[[k, 1, 2]]]},
      {k, 1, Length[cp]}]]}, PlotRange → 10];
Show[plot1, plot2, ImageSize → {250}]
```

Out[388]=



Example 14.36. Let $f(x, y) = 2x^2 - 3xy - x + y + y^2$ and let R be the rectangle in the xy-plane whose vertices are at (0,0), (2,0), (2,2), and (0,2).
a) Find all relative extreme values of f inside R.
b) Find the maximum and minimum values of f on R.

Solution:

In[389]:=
```
Clear[f, x, y, disc]
f[x_, y_] = 2 x^2 - 3 x * y - x + y + y^2 + 5
```

Out[390]= $5 - x + 2 x^2 + y - 3 x y + y^2$

In[391]:=
```
Solve[{∂x f[x, y] == 0, ∂y f[x, y] == 0}, {x, y}]
```

Out[391]= {{x → 1, y → 1}}

In[392]:=
```
disc[x_, y_] = ∂x,x f[x, y] * ∂y,y f[x, y] - (∂x,y f[x, y])^2
```

Out[392]= -1

In[393]:=
```
∂x,x f[x, y] /. {x -> 1, y -> 1}
disc[x, y] /. {x -> 1, y -> 1}
```

Out[393]= 4

Out[394]= -1

Thus, (1, 1) is the local minimum point of f inside R and its local minimum value is $f(1, 1) = 5$. Next, we find the extreme values of f on the boundary of the rectangle. This is done by considering f as a function of one variable corresponding to each side of R. Let $f_1 = f(x, 0)$, $f_2 = f(x, 2)$, for x between 0 and 2, and $f_3 = f(0, y)$ and $f_4 = f(2, y)$, for y between 0 and 2. We now proceed as follows:

In[395]:=
```
Clear[f1, f2, f3, f4]
f1 = f[x, 0]
f2 = f[x, 2]
f3 = f[0, y]
f4 = f[2, y]
```

Out[396]= $5 - x + 2 x^2$

Out[397]= $11 - 7 x + 2 x^2$

Out[398]= $5 + y + y^2$

Out[399]= $11 - 5 y + y^2$

In[400]:=
```
Solve[D[f1, x] == 0 ]
```

Out[400]= $\left\{\left\{x \to \frac{1}{4}\right\}\right\}$

In[401]:=
```
Solve[D[f2, x] == 0 ]
```

Out[401]= $\left\{\left\{x \to \frac{7}{4}\right\}\right\}$

In[402]:=
```
Solve[D[f3, y] == 0 ]
```

Out[402]= $\left\{\left\{y \to -\frac{1}{2}\right\}\right\}$

In[403]:=
```
Solve[D[f4, y] == 0 ]
```

Out[403]= $\left\{\left\{y \to \frac{5}{2}\right\}\right\}$

Thus, points on the boundary of R that are critical points of f are $\left(\frac{1}{4},\ 0\right)$ and $\left(\frac{7}{4},\ 2\right)$. Observe that the points $(0,\ -1/2)$ and $\left(2,\ \frac{5}{2}\right)$ are outside the rectangle R. The four vertices of R at (0,0), (2,0), (0,2) and (2,2) are also critical points. Can you explain why? We now evaluate f at each of these points and at (1, 1) (the relative minimum point found earlier) using the substitution command and compare the results.

In[404]:=
```
f[x, y] /. {{x -> 1/4, y -> 0}, {x -> 7/4, y -> 2},
    {x -> 0, y -> 0}, {x -> 2, y -> 0},
  {x -> 0, y -> 2}, {x -> 2, y -> 2}, {x -> 1, y -> 1}}
```

Out[404]= $\left\{\frac{39}{8},\ \frac{39}{8},\ 5,\ 11,\ 11,\ 5,\ 5\right\}$

Thus, the minimum value of f is 39/8, which occurs at (1/4, 0) and also at (7/4, 2). The maximum value of f is 6, which is attained at (2, 0) and also at (0, 2). Here is the graph of f over the rectangle R.

In[405]:=
```
Clear[plot1, plot2, plot3]
plot1 = Plot3D[{f[x, y], 0}, {x, 0, 2},
   {y, 0, 2}, PlotStyle → {Green, Blue}, PlotRange → All];
plot2 = Graphics3D[{PointSize[Large], Red ,
    Point[{ {1 / 4, 0, f[1 / 4, 0]}, {7 / 4, 2, f[7 / 4, 2]} }]}, PlotRange → All ];
plot3 = Graphics3D[{PointSize[Large], Black ,
    Point[{ {2, 0, f[2, 0]}, {0, 2, f[2, 0]}}]}, PlotRange → All ];
Show[plot1, plot2 , plot3, ImageSize → {250}, ImagePadding → {{15, 15}, {15, 15}}]
```

Out[409]=



■ Exercises

1. Let $f(x,\ y) = x^4 - 4\,x\,y + 2\,y^2$.

a. Find all critical points of f.

b. Use the second derivative test to classify the critical points as local minimum, local maximum, saddle point, or neither.

c. Plot the graph of f and the local extreme points and saddle points, if any.

2. Let $f(x,\ y) = \ (x + y)\ln\left(x^2 + y^2\right)$, for $(x,\ y) \neq (0,\ 0)$.

a. Find all critical points of f.

b. Use the second derivative test to classify the critical points as local minimum, local maximum, saddle point, or neither.

c. Plot the graph of f and the local extreme points and saddle points, if any.

3. Let $f(x, y) = 2x^2 - 3xy - x + y + y^2$ and let R be the rectangle in the xy-plane whose vertices are at (0, 0), (2, 0), (2, 2), and (0, 2).
a. Find all relative extreme values of f inside R.
b. Find the maximum and minimum values of f on R.
c. Plot the graph of f and the local extreme points and saddle points, if any.

14.8 Lagrange Multipliers

Students should read Section 14.8 of Rogawski's *Calculus* [1] for a detailed discussion of the material presented in this section.

Example 14.37. Let $f(x, y) = x\,y$ and $g(x, y) = x^2 + y^2 - 4$.
a) Plot the level curves of f and g as well as their gradient vectors.
b) Find the maximum and minimum values of f subject to the constraint $g(x, y) = 0$.

Solution:
a) We will define f and g and compute their gradients. Recall that we need to evaluate the command **Needs["`VectorFieldPlots`"]** before we plot the gradient fields.

In[410]:=
```
Clear[f, g, fx, fy, gx, gy, x, y]
f[x_, y_] = 2 x + 3 y
g[x_, y_] = x^2 + y^2 - 4
fx = D[f[x, y], x]
fy = D[f[x, y], y]
gx = D[g[x, y], x]
gy = D[g[x, y], y]
```

Out[411]= $2x + 3y$

Out[412]= $-4 + x^2 + y^2$

Out[413]= 2

Out[414]= 3

Out[415]= $2x$

Out[416]= $2y$

In[417]:=
```
Needs["VectorFieldPlots`"]
```

In[418]:=
```
Clear[plot1, plot2, plot3, plot4]
plot1 = ContourPlot[x^2 + y^2 - 4, {x, -2, 2}, {y, -2, 2},
   Frame → False, Axes → True, ContourShading → False, PlotRange → All];
plot2 = ContourPlot[2 x + 3 y, {x, -2, 2}, {y, -2, 2}, Frame → False,
   Axes → True, ContourShading → False, PlotRange → All];
plot3 = VectorFieldPlot[{fx, fy}, {x, -2, 2}, {y, -2, 2},
   Axes → True, Frame → False, ColorFunction → Hue];
plot4 = VectorFieldPlot[{gx, gy}, {x, -2, 2}, {y, -2, 2},
   Axes → True, Frame → False, ColorFunction → Hue];
Show[plot1, plot2, plot3, plot4, ImageSize → {250}]
```

Out[423]=



b) Let us use l for λ. To solve $\nabla f = l\, \nabla g$ we compute

In[424]:=
```
Solve[{fx == l gx, fy == l gy, g[x, y] == 0}]
```

Out[424]= $\left\{\left\{l \to -\frac{\sqrt{13}}{4},\ x \to -\frac{4}{\sqrt{13}},\ y \to -\frac{6}{\sqrt{13}}\right\},\ \left\{l \to \frac{\sqrt{13}}{4},\ x \to \frac{4}{\sqrt{13}},\ y \to \frac{6}{\sqrt{13}}\right\}\right\}$

Thus, $\left(-\frac{4}{\sqrt{13}}, -\frac{6}{\sqrt{13}}\right)$ and $\left(\frac{4}{\sqrt{13}}, \frac{6}{\sqrt{13}}\right)$ are the critical points. We evaluate f at these points to determine the absolute maximum and the absolute minimum of f on the graph of $g(x, y) = 0$.

In[425]:= $f\left[-\frac{4}{\sqrt{13}}, -\frac{6}{\sqrt{13}}\right]$

$f\left[\frac{4}{\sqrt{13}}, \frac{6}{\sqrt{13}}\right]$

Out[425]= $-2\sqrt{13}$

Out[426]= $2\sqrt{13}$

Hence, f attains its absolute minimum value of $-2\sqrt{13}$ at $\left(-\frac{4}{\sqrt{13}}, -\frac{6}{\sqrt{13}}\right)$ and absolute maximum value of $-2\sqrt{13}$ at $\left(\frac{4}{\sqrt{13}}, \frac{6}{\sqrt{13}}\right)$.

Here is a combined plot of the gradients of f (in black) and g (in red) at the critical points.

In[427]:=
```
Clear[plot1, plot2, plot3, plot4, plot5, plot6]
plot1 = ContourPlot[g[x, y], {x, -3, 3}, {y, -3, 3},
   Contours → {0}, Frame → False, Axes → True, ContourShading → False];
```

$$\texttt{plot2 = ListPlot}\left[\left\{\left\{-\frac{4}{\sqrt{13}}, -\frac{6}{\sqrt{13}}\right\}, \left\{\frac{4}{\sqrt{13}}, \frac{6}{\sqrt{13}}\right\}\right\}\right];$$

In[430]:=

$$\texttt{plot3 = Graphics}\Big[\texttt{Arrow}\Big[\left\{\left\{-\frac{4}{\sqrt{13}}, -\frac{6}{\sqrt{13}}\right\}, \left\{-\frac{4}{\sqrt{13}}, -\frac{6}{\sqrt{13}}\right\} + \{\texttt{fx, fy}\} \;/.\; \left\{\texttt{x ->} \frac{-4}{\sqrt{13}}, \texttt{y ->} \frac{-6}{\sqrt{13}}\right\}\right\}\Big]\Big];$$

In[431]:=

$$\texttt{plot4 = Graphics}\Big[\texttt{Arrow}\Big[\left\{\left\{\frac{4}{\sqrt{13}}, \frac{6}{\sqrt{13}}\right\}, \left\{\frac{4}{\sqrt{13}}, \frac{6}{\sqrt{13}}\right\} + \{\texttt{fx, fy}\} \;/.\; \left\{\texttt{x ->} \frac{4}{\sqrt{13}}, \texttt{y ->} \frac{6}{\sqrt{13}}\right\}\right\}\Big]\Big];$$

In[432]:=

$$\texttt{plot5 = Graphics}\Big[\Big\{\texttt{Red, Arrow}\Big[\left\{\left\{-\frac{4}{\sqrt{13}}, -\frac{6}{\sqrt{13}}\right\}, \left\{-\frac{4}{\sqrt{13}}, -\frac{6}{\sqrt{13}}\right\} + \{\texttt{gx, gy}\} \;/.\; \left\{\texttt{x ->} \frac{-4}{\sqrt{13}}, \texttt{y ->} \frac{-6}{\sqrt{13}}\right\}\right\}\Big]\Big\}\Big];$$

In[433]:=

$$\texttt{plot6 = Graphics}\Big[\Big\{\texttt{Red, Arrow}\Big[\left\{\left\{\frac{4}{\sqrt{13}}, \frac{6}{\sqrt{13}}\right\}, \left\{\frac{4}{\sqrt{13}}, \frac{6}{\sqrt{13}}\right\} + \{\texttt{gx, gy}\} \;/.\; \left\{\texttt{x ->} \frac{4}{\sqrt{13}}, \texttt{y ->} \frac{6}{\sqrt{13}}\right\}\right\}\Big]\Big\}\Big];$$

In[434]:=
```
Show[plot1, plot2, plot3, plot4, plot5, plot6,
 PlotRange → All, AspectRatio → Automatic, ImageSize → {250}]
```

Out[434]=



■ Exercises

1. Let $f(x, y) = 4x^2 + 9y^2$ and $g(x, y) = xy - 4$.
a. Plot the level curves of f and g as well as their gradient vectors.
b. Find the maximum and minimum values of f subject to $g(x, y) = 0$.

2. Find the maximum and minimum values of $f(x, y, z) = x^3 - 3y^2 + 4z$ subject to the constraint $g(x, y, z) = x + yz - 4 = 0$.

3. Find the maximum area of a rectangle that can be inscribed in the ellipse $\frac{x^2}{a^2} + \frac{y^2}{b^2} = 1$.

4. Find the maximum volume of a box that can be inscribed in the sphere $x^2 + y^2 + z^2 = 4$.